5-15-2014

Rainich Geometrization of Real Massless Scalar Fields

Dionisios Krongos

Utah State University

Rainich geometrization of real massless scalar fields

Dionisios Krongos
Physics 4900 Report
Department of Physics
Utah State University
Logan, Utah, U.S.A.
May 15, 2014

Abstract

Rainich geometrization is the process of eliminating the source from Einstein's field equations and thus expressing the equations solely with geometric quantities. This report briefly covers the theory, due to Kuchar, involved in the Rainich geometrization of a real massless scalar field with no cosmological constant. The theory contains the conditions which the Ricci tensor must satisfy such that the spacetime permits the scalar field and also explains the method used to reconstruct the field. Two procedures are written which automate this process and they are used extensively through the rest of the paper to both verify existing solutions, such as Fisher's (alternatively, the JNW solution) and Xanthopoulos', and to compute other solutions from generic spacetimes.

Theory

Rainich geometrization of a scalar field is the process of finding conditions on the metric tensor g_{ik} such that from the Einstein field equations the scalar field can be found. The process for determining the scalar field directly through the metric in this worksheet is by the conditions that K. Kuchar found [1]. In the case of the massless scalar field, the Einstein scalar field equations are

$$G_{ik} - \frac{1}{2} G\, g_{ik} = R_{ik} = \psi_{,\,i}\, \psi_{,\,k} \tag{1}$$

Where G_{ik}, R_{ik}, and ψ_i are the Einstein tensor, Ricci tensor, and scalar field, respectively. From here these conditions on the metric tensor g_{ik} will be referred to as the Kuchar conditions. The Kuchar conditions, which involve the Ricci tensor R_{ik} and the covariant derivative ; , are given by

$$R_{i[k}R_{l]m} = R_{ik}R_{lm} - R_{il}R_{km} = 0, \tag{2}$$

which checks whether or not the Ricci tensor can be expressed as a product of vectors, and

$$R_{ik}\, R_{l[m;\,n]} + R_{il}\, R_{k[m;\,n]} - R_{i[m}\, R_{kl;\,n]} = 0, \tag{3}$$

which checks whether or not the vectors are the gradient of a scalar. When these two conditions are satisfied, they guarentee the existence of a scalar field ψ at all non-singular points. The scalar field can be obtained through the following process. First by creating the dual vector

$$\omega = \sum_i \sqrt{R_{ii}}\, dx^i \tag{4}$$

and provided that Kuchar's conditions are satisfied we will have

$$\mathrm{d}\omega = 0. \tag{5}$$

Now we can set up the system of differential equations

$$\mathrm{d}\psi = \omega \tag{6}$$

and solutions to the above yield the massless scalar field which was determined solely from the metric tensor g_{ik}.

Code

```
> restart;
> with(DifferentialGeometry): with(Tensor): with(Tools): with
  (PDETools):
```

The code is included in the appendix.

The code for the procedure which will check if a given metric permits a massless scalar field. The command **KucharConditions** accepts a metric tensor as input and optional commands which alter the output. If only a metric is passed into the procedure, then it will generate the Ricci tensor to see if it can be expressed as a product of vectors, Eq. (2), and those vectors are the gradient of a scalar, Eq. (3). If those two conditions are satisfied then the procedure will display *true*, and if the conditions are not satisfied the procedure will display *false*. Alternatively, the user can request the two tensors generated by the metric which satisfy Kuchar's two conditions. They can get these tensors by passing the metric into **KucharConditions** as well as output = "tensor".

KucharConditions

```
KucharConditions := proc(g0, {output := "TF"})
```

The code for the procedure which will accept a metric that satisfies Kuchar's conditions and will generate the massless scalar field. The command **KucharScalarField** accepts a metric tensor as the input and will output solutions for the scalar field. The procedure will create a dual vector made from components of the Ricci tensor and will set up a differential equation whose solutions are the desired scalar field.

KucharScalarField

```
KucharScalarField := proc(g0)
```

Example 1: Checking Kuchar's Conditions on the Fisher Solutions

The Fisher solution [2] is a static spherically symmetric spacetime with a massless scalar field. In Xanthopoulos' [3] paper he gives metrics in arbitrary dimensons (provided that the dimension of the spacetime is greater than or equal to four) which permit a massless scalar field. We will check the four and five dimensional cases with Kuchar's conditions.

Four-Dimensional Case

```
> with(DifferentialGeometry): with(Tensor): with(Tools):
> DGsetup([t, r, theta, phi], M);
```

frame name: M **(4.1.1)**

The metric takes on the following form with the functions $f1$ and $h1$ defined below

```
> g := evalDG( -exp(f1)*dt &t dt + exp(-h1)*( dr &t dr + r^2 * ( dtheta &t dtheta +  sin(theta)^2 * dphi &t dphi)));
```

$$g := -e^{f1}\, dt \otimes dt + e^{-h1}\, dr \otimes dr + e^{-h1}\, r^2\, d\theta \otimes d\theta + e^{-h1}\, r^2 \sin(\theta)^2\, d\phi \otimes d\phi \quad \textbf{(4.1.2)}$$

```
M > unprotect(gamma);
```

The general form of the functions $f1$ and $h1$ with dimension left unspecified. The term γ has the range of $0 \leq \gamma^2 \leq 1$, $r0$ is a parameter, and d is the dimension of the spacetime.

```
M > f1 := evalDG( ln( ((r^(d - 3) - r0^(d-3))/(r^(d - 3) + r0^(d- 3)))^(2*gamma)));
```

$$f1 := \ln\left(\left(\frac{r^{d-3} - r0^{d-3}}{r^{d-3} + r0^{d-3}}\right)^{2\gamma}\right) \quad \textbf{(4.1.3)}$$

```
M > h1 := evalDG( -ln( ( 1 - r0^(2*d - 6)/r^(2*d - 6))^(2/(d - 3)) * ((r^(d - 3) - r0^(d - 3))/(r^(d-3) + r0^(d-3)))^(-2*gamma/(d-3))));
```

(4.1.4)

$$h1 := -\ln\left(\left(1 - \frac{r0^{2d-6}}{r^{2d-6}}\right)^{\frac{2}{d-3}} \left(\frac{r^{d-3} - r0^{d-3}}{r^{d-3} + r0^{d-3}}\right)^{-\frac{2\gamma}{d-3}}\right) \quad \mathbf{(4.1.4)}$$

Setting the dimension *d* to four leads to

```
M > h4 := eval(h1, d = 4);
```

$$h4 := -\ln\left(\left(1 - \frac{r0^2}{r^2}\right)^2 \left(\frac{r-r0}{r+r0}\right)^{-2\gamma}\right) \quad \mathbf{(4.1.5)}$$

```
M > f4 := eval(f1, d = 4);
```

$$f4 := \ln\left(\left(\frac{r-r0}{r+r0}\right)^{2\gamma}\right) \quad \mathbf{(4.1.6)}$$

```
M > g4 := eval(g, [f1 = f4, h1 = h4, d = 4]);
```

$$g4 := -\left(\frac{r-r0}{r+r0}\right)^{2\gamma} dt \otimes dt + \left(1 - \frac{r0^2}{r^2}\right)^2 \left(\frac{r-r0}{r+r0}\right)^{-2\gamma} dr \otimes dr + \left(1 - \frac{r0^2}{r^2}\right)^2 \left(\frac{r-r0}{r+r0}\right)^{-2\gamma} r^2\, d\theta \otimes d\theta + \left(1 - \frac{r0^2}{r^2}\right)^2 \left(\frac{r-r0}{r+r0}\right)^{-2\gamma} r^2 \sin(\theta)^2\, d\phi \otimes d\phi \quad \mathbf{(4.1.7)}$$

Check the command **KucharConditions** to see if there is a massless scalar field which could give rise to this metric.

```
M > KucharConditions(g4);
```

$$true \quad \mathbf{(4.1.8)}$$

Now that we know that Kuchar's conditions on the metric are satisfied we can see what scalar field is associated with this metric.

```
M > KucharScalarField(g4);
```

$$\left\{(-\ln(r-r0) + \ln(r+r0))\sqrt{-2\gamma^2+2} + \_C1,\ (\ln(r-r0) - \ln(r+r0))\sqrt{-2\gamma^2+2} + \_C1\right\} \quad \mathbf{(4.1.9)}$$

We now have a scalar field and can check to see if the Einstein-Klein-Gordon equations are satisfied. First we'll assign one of the fields to a Maple variable.

```
M > PHI4 := KucharScalarField(g4)[1];
```

$$PHI4 := (-\ln(r-r0) + \ln(r+r0))\sqrt{-2\gamma^2+2} + \_C1 \quad \mathbf{(4.1.10)}$$

With the scalar field and the metric we can generate the Einstein tensor and the energy-momentum tensor.

```
M > G4 := EinsteinTensor(g4);
```

(4.1.11)

$$G4 := -\frac{4\,r^4\,r0^2\,(\gamma^2-1)}{(r^2-r0^2)^4}\,\partial_t\otimes\partial_t - \frac{4\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^8\,r0^2\,(\gamma^2-1)}{(r^2-r0^2)^6}\,\partial_r\otimes\partial_r + \frac{4\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^6\,r0^2\,(\gamma^2-1)}{(r^2-r0^2)^6}\,\partial_\theta\otimes\partial_\theta + \frac{4\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^6\,r0^2\,(\gamma^2-1)}{(r^2-r0^2)^6\,\sin(\theta)^2}\,\partial_\phi\otimes\partial_\phi \tag{4.1.11}$$

```
M > T4 := eval( EnergyMomentumTensor("Scalar", g4, PHI4), _m =
    0);
```

$$T4 := -\frac{8\,\gamma^2\,r^4\,r0^2 - 8\,r^4\,r0^2}{2\,(r^2-r0^2)^4}\,\partial_t\otimes\partial_t + \frac{r^4\left(-8\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^4\,\gamma^2\,r0^2 + 8\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^4\,r0^2\right)}{2\,(r^2-r0^2)^6}\,\partial_r\otimes\partial_r + \frac{r^2\left(8\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^4\,\gamma^2\,r0^2 - 8\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^4\,r0^2\right)}{2\,(r^2-r0^2)^6}\,\partial_\theta\otimes\partial_\theta + \frac{r^2\left(8\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^4\,\gamma^2\,r0^2 - 8\left(\frac{r-r0}{r+r0}\right)^{4\gamma} r^4\,r0^2\right)}{2\,(r^2-r0^2)^6\,\sin(\theta)^2}\,\partial_\phi\otimes\partial_\phi \tag{4.1.12}$$

Einstein's equations should now be satisfied.

```
M > evalDG( G4 - T4);
```

$$0\,\partial_t\otimes\partial_t \tag{4.1.13}$$

And the Klein-Gordon equation should be satisfied as well.

```
M > MFE4 := MatterFieldEquations("Scalar", g4, PHI4);
```

$$MFE4 := -\_m^2\left((-\ln(r-r0)+\ln(r+r0))\sqrt{-2\,\gamma^2+2} + \_C1\right) \tag{4.1.14}$$

```
M > eval(MFE4, _m = 0);
```

$$0 \tag{4.1.15}$$

Five-Dimensional Case

```
> DGsetup([t, r, chi, theta, phi], M1);
```

$$\text{frame name: } M1 \tag{4.2.1}$$

The metric in five dimensions is of the form:

```
M > g1 := evalDG( -exp(f1)*dt &t dt + exp(-h1)*( dr &t dr +
    r^2 *( dchi &t dchi + sin(chi)^2* dtheta &t dtheta + sin
    (chi)^2 * sin(theta)^2 * dphi &t dphi)));
```

$$g1 := -\left(-\frac{r0^d r^3 - r^d r0^3}{r^d r0^3 + r0^d r^3}\right)^{2\gamma} dt \otimes dt + \left(-\frac{r0^{2d} r^{-2d+6} - r0^6}{r0^6}\right)^{\frac{2}{d-3}} \left(-\frac{r0^d r^3 - r^d r0^3}{r^d r0^3 + r0^d r^3}\right)^{-\frac{2\gamma}{d-3}} dr \otimes dr + \left(-\frac{r0^{2d} r^{-2d+6} - r0^6}{r0^6}\right)^{\frac{2}{d-3}} \left(-\frac{r0^d r^3 - r^d r0^3}{r^d r0^3 + r0^d r^3}\right)^{-\frac{2\gamma}{d-3}} r^2\, d\chi \otimes d\chi + \left(-\frac{r0^{2d} r^{-2d+6} - r0^6}{r0^6}\right)^{\frac{2}{d-3}} \left(-\frac{r0^d r^3 - r^d r0^3}{r^d r0^3 + r0^d r^3}\right)^{-\frac{2\gamma}{d-3}} r^2 \sin(\chi)^2\, d\theta \otimes d\theta + \left(-\frac{r0^{2d} r^{-2d+6} - r0^6}{r0^6}\right)^{\frac{2}{d-3}} \left(-\frac{r0^d r^3 - r^d r0^3}{r^d r0^3 + r0^d r^3}\right)^{-\frac{2\gamma}{d-3}} r^2 \sin(\chi)^2 \sin(\theta)^2\, d\phi \otimes d\phi \tag{4.2.2}$$

Once again specifying the dimension of the spacetime.

```
M > g5 := eval(g1, d = 5);
```

$$g5 := -\left(-\frac{-r^5 r0^3 + r0^5 r^3}{r^5 r0^3 + r0^5 r^3}\right)^{2\gamma} dt \otimes dt - \frac{\left(\frac{r0^{10}}{r^4} - r0^6\right)\left(-\frac{-r^5 r0^3 + r0^5 r^3}{r^5 r0^3 + r0^5 r^3}\right)^{-\gamma}}{r0^6} dr \otimes dr - \frac{\left(\frac{r0^{10}}{r^4} - r0^6\right)\left(-\frac{-r^5 r0^3 + r0^5 r^3}{r^5 r0^3 + r0^5 r^3}\right)^{-\gamma} r^2}{r0^6} d\chi \otimes d\chi \tag{4.2.3}$$

$$-\frac{\left(\frac{r0^{10}}{r^4}-r0^6\right)\left(-\frac{-r^5\,r0^3+r0^5\,r^3}{r^5\,r0^3+r0^5\,r^3}\right)^{-\gamma} r^2\sin(\chi)^2}{r0^6}\,d\theta\otimes d\theta$$

$$-\frac{\left(\frac{r0^{10}}{r^4}-r0^6\right)\left(-\frac{-r^5\,r0^3+r0^5\,r^3}{r^5\,r0^3+r0^5\,r^3}\right)^{-\gamma} r^2\sin(\chi)^2\sin(\theta)^2}{r0^6}\,d\phi\otimes d\phi$$

This metric should satisfy Kuchar's conditions.

```
M1 > KucharConditions(g5);
```

$$true \tag{4.2.4}$$

And the associated fields are:

```
M1 > KucharScalarField(g5);
```

$$\left\{-\frac{\ln(r^2-r0^2)\sqrt{-6\gamma^2+6}}{2}+\frac{\ln(r^2+r0^2)\sqrt{-6\gamma^2+6}}{2}+\_C1,\ \frac{\ln(r^2-r0^2)\sqrt{-6\gamma^2+6}}{2}-\frac{\ln(r^2+r0^2)\sqrt{-6\gamma^2+6}}{2}+\_C1\right\} \tag{4.2.5}$$

Now to verify the Einstein-Klein-Gordon equations we'll need to assign the scalar field to a Maple variable.

```
M1 > PHI5 := KucharScalarField(g5)[1];
```

$$PHI5 := -\frac{\ln(r^2-r0^2)\sqrt{-6\gamma^2+6}}{2}+\frac{\ln(r^2+r0^2)\sqrt{-6\gamma^2+6}}{2}+\_C1 \tag{4.2.6}$$

The Einstein tensor is:

```
M1 > G5 := EinsteinTensor(g5);
```

$$G5 := -\frac{12\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{-\gamma} r^6\,r0^4\,(\gamma^2-1)}{(r^4-r0^4)^3}\,\partial_t\otimes\partial_t \tag{4.2.7}$$

$$-\frac{12\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^{10}\,r0^4\,(\gamma^2-1)}{(r^4-r0^4)^4}\,\partial_r\otimes\partial_r$$

$$+\frac{12\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^8\,r0^4\,(\gamma^2-1)}{(r^4-r0^4)^4}\,\partial_\chi\otimes\partial_\chi$$

$$+\frac{12\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^8\, r0^4\,(\gamma^2-1)}{(r^4-r0^4)^4\sin(\chi)^2}\,\partial_\theta\otimes\partial_\theta$$

$$+\frac{12\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^8\, r0^4\,(\gamma^2-1)}{(r^4-r0^4)^4\sin(\chi)^2\sin(\theta)^2}\,\partial_\phi\otimes\partial_\phi$$

and the energy-momentum tensor is

```
M1 > T5 := eval( EnergyMomentumTensor("Scalar", g5, PHI5), _m
     = 0);
```

$$T5 := -\frac{48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{-\gamma}\gamma^2 r^6\, r0^4-48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{-\gamma} r^6\, r0^4}{4\,(r^4-r0^4)^3}\,\partial_t\otimes\partial_t \tag{4.2.8}$$

$$+\frac{r^4\left(48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^6\, r0^4-48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma}\gamma^2 r^6\, r0^4\right)}{4\,(r^4-r0^4)^4}\,\partial_r\otimes\partial_r$$

$$+\frac{r^2\left(-48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^6\, r0^4+48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma}\gamma^2 r^6\, r0^4\right)}{4\,(r^4-r0^4)^4}\,\partial_\chi\otimes\partial_\chi$$

$$+\frac{r^2\left(-48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^6\, r0^4+48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma}\gamma^2 r^6\, r0^4\right)}{4\,(r^4-r0^4)^4\sin(\chi)^2}\,\partial_\theta\otimes\partial_\theta$$

$$+\frac{r^2\left(-48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma} r^6\, r0^4+48\left(\frac{r^2-r0^2}{r^2+r0^2}\right)^{2\gamma}\gamma^2 r^6\, r0^4\right)}{4\,(r^4-r0^4)^4\sin(\chi)^2\sin(\theta)^2}\,\partial_\phi\otimes\partial_\phi$$

Now verify that the scalar field was indeed a solution. Einstein's equations will vanish.

```
M1 > evalDG( G5 - T5);
```

$$0\,\partial_t\otimes\partial_t \tag{4.2.9}$$

As will the Klein-Gordon equation.

```
M1 > MFE5 := MatterFieldEquations("Scalar", g5, PHI5);
```

$$MFE5 := -\_m^2\left(-\frac{\ln(r^2-r0^2)\sqrt{-6\gamma^2+6}}{2}+\frac{\ln(r^2+r0^2)\sqrt{-6\gamma^2+6}}{2}\right. \tag{4.2.10}$$

$+ \_C1\Big)$

```
M1 > eval( MFE5, _m = 0);
```

$$0 \tag{4.2.11}$$

Example 2: Solving Kuchar's Conditions for the Spherically Symmetric Metric

In this example we'll start with a static spherically symmetric ansatz for a metric and find functions $f(r)$ and $h(r)$, which the metric depends upon, that will satisfy Einstein's equations and the Klein-Gordon equation.

```
M1 > DGsetup([t, r, theta, phi], M2);
```

$$\textit{frame name: M2} \tag{5.1}$$

The general static spherically symmetric spacetime with the arbitrary functions $f(r)$ and $h(r)$ will be defined as

```
M2 > g2 := evalDG( -exp(f(r))*dt &t dt + exp(-h(r))*( dr &t dr +
     r^2 * ( dtheta &t dtheta +  sin(theta)^2 * dphi &t dphi)));
```

$$g2 := -\mathrm{e}^{f(r)}\, dt \otimes dt + \mathrm{e}^{-h(r)}\, dr \otimes dr + \mathrm{e}^{-h(r)}\, r^2\, d\theta \otimes d\theta + \mathrm{e}^{-h(r)}\, r^2 \sin(\theta)^2\, d\phi \otimes d\phi \tag{5.2}$$

Now since this metric has two arbitrary functions in it Kuchar's conditions should fail because the Ricci tensor will be too general.

```
M2 > KucharConditions(g2);
```

$$\textit{false} \tag{5.3}$$

The metric didn't satisfy Kuchar's conditions for Rainich geometrization, as expected. Now we can use the alternative output of the procedure **KucharConditions** to get the equations which the Ricci tensor must satisfy to permit Rainich geometrization. Using **KucharConditions**, the two conditions can be retrieved so we can solve for the arbitrary functions which will satisfy Kuchar's conditons. The output A and B are tensors which are quite large so the output is supressed.

```
M2 > A, B := KucharConditions(g2, output = "tensor"):
```

One of the above tensors is the condition on the Ricci tensor where the Ricci tensor will be written as a product of vectors, and the other is the condition on the Ricci tensor such that the product of vectors which make up the Ricci tensor are the gradient of a scalar. Now we can take the coefficients of A and B to set up a system of differential equations

```
M2 > eq := DGinfo(A, "CoefficientSet") union DGinfo(B,
     "CoefficientSet"):
```

With the system of differential equations assigned to *eq* we can now solve for $f(r)$ and $h(r)$.

```
M2 > A:
M2 > eq:
M2 > sol2 := Solve(eq);
```

$$sol2 := \left\{f(r) = -\_C2\ln(2) + \_C2\ln\left(-\frac{\_C2\,(\_C4+r)}{\_C4-r}\right) - \_C3,\ h(r) = -\_C2\ln(2) + \_C2\ln\left(-\frac{\_C2\,(\_C4+r)}{\_C4-r}\right) + 2\ln(2) + 4\ln(r) + 2\ln\left(\frac{\_C2\,\_C4}{(\_C4-r)\,(\_C4+r)}\right) + \_C1 - \_C3\right\}, \left\{f(r) = \_C2\ln\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right) + \_C2\ln(2) - \_C3,\ h(r) = \_C2\ln\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right) + \_C2\ln(2) + 2\ln(2) + 4\ln(r) + 2\ln\left(\frac{\_C2\,\_C4}{(\_C4-r)\,(\_C4+r)}\right) + \_C1 - \_C3\right\}, \left\{f(r) = \_C1,\ h(r) = 4\ln(2) - 2\ln\left(\frac{-2\,\_C3\,r^2 + \_C2}{r^2}\right)\right\} \quad \textbf{(5.4)}$$

With solutions for both $f(r)$ and $h(r)$ we can now see if Kuchar's Conditions are satisfied. Defining a new metric $g22$ with the second solution:

```
M2 > g22 := eval(g2, sol2[2]);
```

$$g22 := -\mathrm{e}^{\_C2\ln\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right) + \_C2\ln(2) - \_C3}\, dt \otimes dt + \mathrm{e}^{-\_C2\ln\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right) - \_C2\ln(2) - 2\ln(2) - 4\ln(r) - 2\ln\left(\frac{\_C2\,\_C4}{(\_C4-r)\,(\_C4+r)}\right) - \_C1 + \_C3}\, dr \otimes dr + \mathrm{e}^{-\_C2\ln\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right) - \_C2\ln(2) - 2\ln(2) - 4\ln(r) - 2\ln\left(\frac{\_C2\,\_C4}{(\_C4-r)\,(\_C4+r)}\right) - \_C1 + \_C3}\, r^2\, d\theta \otimes d\theta + \mathrm{e}^{-\_C2\ln\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right) - \_C2\ln(2) - 2\ln(2) - 4\ln(r) - 2\ln\left(\frac{\_C2\,\_C4}{(\_C4-r)\,(\_C4+r)}\right) - \_C1 + \_C3}\, r^2 \sin(\theta)^2\, d\phi \otimes d\phi \quad \textbf{(5.5)}$$

Checking **KucharConditions** should return true.

```
M2 > KucharConditions(g22);
```

$$true \quad \textbf{(5.6)}$$

Since Kuchar's conditions are satisfied we can now solve for the masslesss scalar field by passing the the metric into the procedure **KucharScalarField**

```
M2 > KucharScalarField(g22);
```

(5.7)

$$\frac{\sqrt{2}\,\sqrt{-\_C2^2+4}\,\ln(\_C4+r)}{2}-\frac{\sqrt{2}\,\sqrt{-\_C2^2+4}\,\ln(-\_C4+r)}{2}+\_C5 \quad \textbf{(5.7)}$$

The procedure **KucharScalarField** found us a scalar field, but since this is a solution for a static spherically symmetric metric, as is Fisher's solution, we'll now see if the two are equivalent. The solution that **KucharScalarField** found is *PHI2*

```
M2 > PHI2 := KucharScalarField(g22);
```

$$PHI2 := \frac{\sqrt{2}\,\sqrt{-\_C2^2+4}\,\ln(\_C4+r)}{2}-\frac{\sqrt{2}\,\sqrt{-\_C2^2+4}\,\ln(-\_C4+r)}{2}+\_C5 \quad \textbf{(5.8)}$$

and the solution found by Fisher is *PHI4*

```
M > PHI4;
```

$$(-\ln(r-r0)+\ln(r+r0))\,\sqrt{-2\,\gamma^2+2}+\_C1 \quad \textbf{(5.9)}$$

With the two solutions we can evaluate the fields with different constants to see if they are equivalent solutions.

```
M2 > S2 := simplify( eval(PHI2, {_C5 = _C1, _C4 = r0, _C2 = 2*
     gamma}));
```

$$S2 := \sqrt{2}\,\sqrt{-\gamma^2+1}\,\ln(r+r0)-\sqrt{2}\,\sqrt{-\gamma^2+1}\,\ln(r-r0)+\_C1 \quad \textbf{(5.10)}$$

```
M2 > simplify( S2 - PHI4);
```

$$0 \quad \textbf{(5.11)}$$

This remainder is unecessary since the solution was found to be equivalent to Fisher's solution, however the validity of the solution can be checked in only a few steps.

```
M2 > G2 := EinsteinTensor(g22);
```

$$G2 := -\frac{4\,r^4\,\_C2^2\,\_C4^4\,e^{-\_C1}\,(\_C2^2-4)}{(\_C4-r)^2\,(\_C4+r)^2\,(\_C4^2-r^2)^2}\,\partial_t\otimes\partial_t \quad \textbf{(5.12)}$$
$$-\frac{16\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,r^8\,\_C2^4\,\_C4^6\,e^{2\,\_C1-2\,\_C3}\,(\_C2^2-4)}{(\_C4-r)^4\,(\_C4+r)^4\,(\_C4^2-r^2)^2}\,\partial_r\otimes\partial_r$$
$$+\frac{16\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,r^6\,\_C2^4\,\_C4^6\,e^{2\,\_C1-2\,\_C3}\,(\_C2^2-4)}{(\_C4-r)^4\,(\_C4+r)^4\,(\_C4^2-r^2)^2}\,\partial_\theta\otimes\partial_\theta$$
$$+\frac{16\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,r^6\,\_C2^4\,\_C4^6\,e^{2\,\_C1-2\,\_C3}\,(\_C2^2-4)}{(\_C4-r)^4\,(\_C4+r)^4\,(\_C4^2-r^2)^2\,\sin(\theta)^2}\,\partial_\phi\otimes\partial_\phi$$

Generating the energy-momentum tensor with mass set to zero:

```
M2 > T2 := eval(EnergyMomentumTensor("Scalar", g22, PHI2), _m =
     0);
```

$$T2 := -\frac{e^{-\_C3}\left(-64\,e^{-\_C1-\_C3}\,\_C4^4\,\_C2^2\,r^4+16\,e^{-\_C1-\_C3}\,\_C2^4\,\_C4^4\,r^4\right)}{4\,(\_C4-r)^2\,(\_C4+r)^2\,(\_C4^2-r^2)^2}\,\partial_t\otimes\partial_t$$

$$+\frac{1}{(\_C4-r)^4\,(\_C4+r)^4\,(\_C4^2-r^2)^2}\Bigg(\_C4^2\,\_C2^2\,r^4\,e^{-\_C1-\_C3}\Bigg(64$$

$$\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,e^{-\_C1-\_C3}\,\_C4^4\,\_C2^2\,r^4$$

$$-16\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,e^{-\_C1-\_C3}\,\_C2^4\,\_C4^4\,r^4\Bigg)\Bigg)\,\partial_r\otimes\partial_r$$

$$+\frac{1}{(\_C4-r)^4\,(\_C4+r)^4\,(\_C4^2-r^2)^2}\Bigg(r^2\,\_C2^2\,\_C4^2\,e^{-\_C1-\_C3}\Bigg($$

$$-64\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,e^{-\_C1-\_C3}\,\_C4^4\,\_C2^2\,r^4$$

$$+16\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,e^{-\_C1-\_C3}\,\_C2^4\,\_C4^4\,r^4\Bigg)\Bigg)\,\partial_\theta\otimes\partial_\theta$$

$$+\Bigg(r^2\,\_C2^2\,\_C4^2\,e^{-\_C1-\_C3}\Bigg(-64\left(\frac{\_C4+r}{\_C2\,(\_C4-r)}\right)^{2\,\_C2}4^{-\_C2}\,e^{-\_C1-\_C3}\,\_C4^4\,\_C2^2\,r^4+16\Bigg(\frac{\quad}{\_C2}$$

$$-r^2)^2\,\sin(\theta)^2\Bigg)\,\partial_\phi\otimes\partial_\phi$$

Now for the Einstein field equations we have:

```
M2 > evalDG( G2 - T2);
```

$$0\,\partial_t\otimes\partial_t \tag{5.14}$$

And for the Klein-Gordon equation we have:

```
M2 > MatterFieldEquations("Scalar", g22, PHI2);
```

$$-\_m^2\left(\frac{\sqrt{2}\,\sqrt{-\_C2^2+4}\,\ln(\_C4+r)}{2}-\frac{\sqrt{2}\,\sqrt{-\_C2^2+4}\,\ln(-\_C4+r)}{2}+\_C5\right) \tag{5.15}$$

Which clearly vanishes for the massless case ($\_m = 0$)

```
M2 > eval(MatterFieldEquations("Scalar", g22, PHI2), _m = 0);
```

$$0 \tag{5.16}$$

Example 3: 2 + 1 Spacetime, Solving Kuchar's Conditions for

the General Static Rotationally Symmetric Metric

In this example we'll look at a three dimensional metric.

```
> DGsetup([t, r, theta], M3);
```

$$\text{frame name: } M3 \tag{6.1}$$

Let $g3$ be the static rotationally symmetric metric.

```
M3 > g3 := evalDG( -f(r)^2 *dt &t dt + h(r)^2 * dr &t dr + r^2 *
     dtheta &t dtheta);
```

$$g3 := -f(r)^2\, dt \otimes dt + h(r)^2\, dr \otimes dr + r^2\, d\theta \otimes d\theta \tag{6.2}$$

Just to make sure that this metric has the right symmetries we'll check the Killing vectors.

```
M3 > KillingVectors(g3);
```

$$\left[\partial_{\theta}, -\partial_{t}\right] \tag{6.3}$$

Once again, this metric probably won't satisfy Kuchar's conditions.

```
M3 > KucharConditions(g3);
```

$$\mathit{false} \tag{6.4}$$

So now we can solve for the functions in the metric.

```
M3 > A1, B1 := KucharConditions(g3, output = "tensor"):
```

Patch together the equations

```
M3 > eq1 := DGinfo(A1, "CoefficientSet") union DGinfo(B1,
     "CoefficientSet"):
M3 > sol3 := Solve(eq1);
```

$$sol3 := \left\{f(r) = \sqrt{2\,\_C2 \ln(r) + 2\,\_C3},\, h(r) = \frac{\_C1}{\sqrt{2\,\_C2 \ln(r) + 2\,\_C3}}\right\}, \left\{f(r) = -\sqrt{2\,\_C2 \ln(r) + 2\,\_C3},\, h(r) = -\frac{\_C1}{\sqrt{2\,\_C2 \ln(r) + 2\,\_C3}}\right\}, \left\{f(r) = \mathrm{e}^{\frac{\_C2\, r^2}{2}}\,\_C3,\, h(r) = \_C1\, \mathrm{e}^{\frac{\_C2\, r^2}{2}}\,\_C3\right\}, \left\{f(r) = r^{-\_C2}\,\_C3,\, h(r) = \_C1\, r^{-\_C2}\,\_C3\right\} \tag{6.5}$$

Let's investigate the third solution.

```
M3 > g33 := eval(g3, sol3[3]);
```

$$g33 := -\left(\mathrm{e}^{\frac{\_C2\, r^2}{2}}\right)^2 \_C3^2\, dt \otimes dt + \_C1^2 \left(\mathrm{e}^{\frac{\_C2\, r^2}{2}}\right)^2 \_C3^2\, dr \otimes dr + r^2\, d\theta \otimes d\theta \tag{6.6}$$

First we make sure it satisfies Kuchar's conditions.

```
M3 > KucharConditions(g33);
```

$$\mathit{true} \tag{6.7}$$

Since it does satisfy Kuchar's conditions we can now create the field associated with the metric.

```
M3 > PHI3 := KucharScalarField(g33);
```

$$PHI3 := \frac{\sqrt{2}\,\sqrt{\_C2}\;t + \_C1\;\_C3}{\_C1} \tag{6.8}$$

As a side note, the scalar field associated with this metric grows in time while the metric does not. Since we have the field and the metric we'll check to see if they solve Einstein's equations. First we have the Einstein tensor,

```
M3 > G3 := EinsteinTensor(g33);
```

$$G3 := \frac{e^{-2\,\_C2\,r^2}\,\_C2}{\_C3^4\,\_C1^2}\,\partial_t \otimes \partial_t + \frac{e^{-2\,\_C2\,r^2}\,\_C2}{\_C3^4\,\_C1^4}\,\partial_r \otimes \partial_r + \frac{e^{-\_C2\,r^2}\,\_C2}{\_C3^2\,\_C1^2\,r^2}\,\partial_\theta \otimes \partial_\theta \tag{6.9}$$

Now we calculate the energy momentum tensor for a massless scalar field.

```
M3 > T3 := eval(EnergyMomentumTensor("Scalar", g33, PHI3), _m =
      0);
```

$$T3 := \frac{\left(e^{-\_C2\,r^2}\right)^2\,\_C2}{\_C3^4\,\_C1^2}\,\partial_t \otimes \partial_t + \frac{\left(e^{-\_C2\,r^2}\right)^2\,\_C2}{\_C3^4\,\_C1^4}\,\partial_r \otimes \partial_r + \frac{e^{-\_C2\,r^2}\,\_C2}{\_C3^2\,\_C1^2\,r^2}\,\partial_\theta \otimes \partial_\theta \tag{6.10}$$

Now we can see that the Einstein scalar field equations are satisfied:

```
M3 > evalDG( G3 - T3);
```

$$0\,\partial_t \otimes \partial_t \tag{6.11}$$

```
M3 > MFE3 := MatterFieldEquations("Scalar", g33, PHI3);
```

$$MFE3 := -\frac{\_m^2\left(\sqrt{2}\,\sqrt{\_C2}\;t + \_C1\;\_C3\right)}{\_C1} \tag{6.12}$$

```
M3 > eval(MFE3, _m = 0);
```

$$0 \tag{6.13}$$

With the solutions found above, the metrics and their associated scalar fields are listed below. While all these metrics solve the Einstein scalar field equations, only the metrics g33 and g34 and their scalar fields are physically valid.

```
M3 > g31 := eval( g3, sol3[1]);
```

$$g31 := -\left|2\,\_C2\ln(r) + 2\,\_C3\right|\,dt \otimes dt + \frac{\_C1^2}{2\,\_C2\ln(r) + 2\,\_C3}\,dr \otimes dr + r^2\,d\theta \otimes d\theta \tag{6.14}$$

```
M3 > KucharConditions(g31);
```

$$true \tag{6.15}$$

```
M3 > KucharScalarField(g31);
```

$$\frac{I\,\theta\sqrt{2}\,\sqrt{\_C2} + \_C1\;\_C3}{\_C1} \tag{6.16}$$

```
M3 > g32 := eval( g3, sol3[2]);
```

$$g32 := -\left|2\,\_C2\ln(r) + 2\,\_C3\right|\,dt \otimes dt + \frac{\_C1^2}{2\,\_C2\ln(r) + 2\,\_C3}\,dr \otimes dr + r^2\,d\theta \otimes d\theta \tag{6.17}$$

```
M3 > KucharConditions(g32);
```

$$true \tag{6.18}$$

```
M3 > KucharScalarField(g32);
```

$$\frac{I\,\theta\sqrt{2}\,\sqrt{\_C2} + \_C1\,\_C3}{\_C1} \tag{6.19}$$

```
M3 > g33 := eval( g3, sol3[3]);
```

$$g33 := -\left(e^{\frac{\_C2\, r^2}{2}}\right)^2 \_C3^2\, dt \otimes dt + \_C1^2 \left(e^{\frac{\_C2\, r^2}{2}}\right)^2 \_C3^2\, dr \otimes dr + r^2\, d\theta \otimes d\theta \tag{6.20}$$

```
M3 > KucharConditions(g33);
```

$$true \tag{6.21}$$

```
M3 > KucharScalarField(g33);
```

$$\frac{\sqrt{2}\,\sqrt{\_C2}\; t + \_C1\,\_C3}{\_C1} \tag{6.22}$$

```
M3 > g34 := eval( g3, sol3[4]);
```

$$g34 := -\left(r^{\_C2}\right)^2 \_C3^2\, dt \otimes dt + \_C1^2 \left(r^{\_C2}\right)^2 \_C3^2\, dr \otimes dr + r^2\, d\theta \otimes d\theta \tag{6.23}$$

```
M3 > KucharConditions(g34);
```

$$true \tag{6.24}$$

```
M3 > KucharScalarField(g34);
```

$$\sqrt{2}\,\sqrt{\_C2}\;\ln(r) + \_C3 \tag{6.25}$$

Conclusion

The Einstein scalar field equations can be geometrized so solutions can be found and verified without using a scalar field. Using Kuchar's methods, the conditions that a metric be associated to a massless scalar field are just two conditions on the Ricci tensor. Using the DifferentialGeometry software I wrote two procedures, one of which will quickly check whether a metric solves the Einstein scalar field equations, and if it doesn't the procedure can return the conditions which the Ricci tensor must satisfy, and the second will solve for the field.

References

1. Kuchar, K., "On the Rainich geometrization of scalar meson fields," *Czech. J. Phys. B* **13** (1963), 551-557.

2. Fisher J. Z., *J. Exp. Th. Phys.* **18** (1948), 636-640.

3. Xanthopoulos, B. C., and Zannias, T., "Einstein gravity coupled to a massless scalar field in arbitrary spacetime dimensions," *Phys. Rev. D Part. Fields.* **40** (1989), 2564-2567.

Appendix

Here is the outline of the code for the procedure **KucharConditions**:

```
> print(KucharConditions);
```

```
proc(g0, {output := "TF"})                                                    (9.1)
    local g, R, RR, RRS, cont, C, covR, p1RR, p3RR, part1, part2, part3, eq20, Z;
    if nargs = 0 then error "expected 1 argument [tensor(metric), keyword(output)]"
    end if;
    g := DifferentialGeometry:-evalDG(g0);
    if not g::'DGmetric' then
        error "expected 1st argument to be a metric tensor.  Received %1",
        DGErrorPrint(g)
    end if;
    R := DifferentialGeometry:-Tensor:-RicciTensor(g);
    RR := DifferentialGeometry:-evalDG(R &t R);
    RRS := DifferentialGeometry:-Tensor:-SymmetrizeIndices(RR, [2, 3],
    "SkewSymmetric");
    RRS := DifferentialGeometry:-Tools:-DGsimplify(RRS);
    C := DifferentialGeometry:-Tensor:-Christoffel(g);
    covR := DifferentialGeometry:-Tensor:-CovariantDerivative(R, C);
    p1RR := DifferentialGeometry:-evalDG(R &t covR);
    part1 := DifferentialGeometry:-Tensor:-SymmetrizeIndices(p1RR, [4, 5],
    "SkewSymmetric");
    part2 := DifferentialGeometry:-Tensor:-RearrangeIndices(part1, [1, 3, 2, 4, 5]);
    p3RR := DifferentialGeometry:-evalDG(R &t covR);
    part3 := DifferentialGeometry:-Tensor:-RearrangeIndices(p3RR, [1, 4, 2, 3, 5]);
    part3 := DifferentialGeometry:-Tensor:-SymmetrizeIndices(part3, [4, 5],
    "SkewSymmetric");
    eq20 := DifferentialGeometry:-evalDG(part1 + part2 - part3);
    eq20 := simplify(eq20, symbolic);
    if output = "TF" then
        Z := DifferentialGeometry:-Tools:-DGinfo(RRS, "CoefficientSet");
        if Z <> {0} then return false end if
    end if;
    if output = "TF" then
        Z := DifferentialGeometry:-Tools:-DGinfo(eq20, "CoefficientSet");
```

```
        if Z <> {0} then return false end if
    end if;
    if output = "TF" then true else RRS, eq20 end if
end proc
```

Here is the outline of the code for the procedure **KucharScalarField**:

```
> print(KucharScalarField);
```

(9.2)

```
proc(g0)
    local g, manifoldName, coordinates, frameForms, frameVectors, numVars, C, m, a, A,
    b, B, eq, phiSol;
    g := DifferentialGeometry:-evalDG(g0);
    manifoldName := DifferentialGeometry:-Tools:-DGinfo("CurrentFrame");
    coordinates := DifferentialGeometry:-Tools:-DGinfo(manifoldName,
    "FrameIndependentVariables");
    frameForms := DifferentialGeometry:-Tools:-DGinfo(manifoldName,
    "FrameBaseForms");
    frameVectors := DifferentialGeometry:-Tools:-DGinfo(manifoldName,
    "FrameBaseVectors");
    numVars := nops(frameForms);
    C := DifferentialGeometry:-Tensor:-Christoffel(g);
    for m to numVars do
        if DifferentialGeometry:-Hook([frameVectors[m], frameVectors[m]],
        DifferentialGeometry:-Tensor:-RicciTensor(g)) <> 0 then
            a[m] := sqrt(DifferentialGeometry:-Hook([frameVectors[m], frameVectors[m
            ]], DifferentialGeometry:-Tensor:-RicciTensor(g)))
        else
            a[m] := 0
        end if
    end do;
    A := DifferentialGeometry:-DGzip(a, frameForms, "plus");
    DifferentialGeometry:-ExteriorDerivative(A);
    B := b(op(coordinates));
    eq := DifferentialGeometry:-Tools:-DGinfo(DifferentialGeometry:-evalDG(convert(A,
    DifferentialGeometry:-Tools:-DGtensor) − DifferentialGeometry:-Tensor:-
    CovariantDerivative(B, C)), "CoefficientSet");
    phiSol := pdsolve(eq);
    if nops([phiSol]) = 1 then
        phiSol := op(simplify(pdsolve(eq), symbolic))
```

```
        else
            phiSol := pdsolve(eq)
        end if
    end proc
```